
	Analyzing data – tips, tricks

and experiences

Session 2 – some important commands



In any programming language there are many of ways to manipulate numbers and texts. It might be pure mathe​matics or just for storage. The examples are just a few of the possible ones that I found very useful. Further information is obtained via the menu [Help], the link Session Commands and the links Alphabetical Command List or Session Commands by Function:
help let                            # Gives all details and examples of the commands.

help set

help tset                           # 'tset' is similar to set but for text strings.

help copy

help concatenate                    # For joining strings.

 Describe
[Stat]>[Basic Statistics]>[Display Descriptive Statistics]
DESCRIBE (or DESC) is used to give a fast overview of a set of data in a numerical form. You can decide exactly which ones of the numerical values to be shown in your session window by opening the menu and tick the ones wanted. Obviously DESC only works for numerical columns: 

desc c1                            # Gives a set of numerical values for data in c1.

desc c1;                           # Describes the data in c1, stratified by column c12. (NB the ';')

by c12.                            # c12 contains either integers or text strings.

 Let
[Calc]>[Calculator]

When you prepare your data set for analysis it is important that it only contains the raw data, not compiled or semi-compiled columns like averages, percentages, etc. Instead you will calculate these values yourself or using the menus to work on the raw data (See it this way – the designers of Minitab did not expect you to do some calculations before starting the analysis. An example: In a control chart for proportions you enter the number of inspected units and the number of failed units. The software then calculates the proportions.):
let k1 = 0                         # Initiates the constant k1.

let k1 = k1 + 1                    # Increases k1 by '1'.

let c5(k1) = "blue"                # Stores 'blue' in cell k1 of c5.

let c1 = k1*(c2 + c3) – 60         # Lets c1 be a function of c2 and c3.

let c11 = c11 – mean(c11)          # Subtracts the mean from every row of c11.
let c2(28) = 2.35                  # Stores 2.35 in row 28 of c2.
let k123 = now()                   # Stores 'now' (e.g. 2006-11-21 07:20:09) in k123.
Not possible:
let k1 = c1            # Not possible to store a column in a constant.

Possible:
let k3 = sum(c34)      # Puts the sum of c34 in k3.

Dangerous:
let c9 = c9 – mean(c9) # Works ok but destroys the original data of c9.

	The menu [Calc]>[Calculator] gives the current time in c1 as '07:08:13' by

let c1 = ctime()

Under 'Functions' a large number of differrent functions are available. 

There are also a number of buttons to create so-called logical expressions:

let c7 = (c4 > 2 and c45 < k1)

If the expression in brackets is true, c7 will contain a '1', otherwise '0'. The bracket contains a 'logical expression'.


Set
[Calc]>[Make…]>[Simple Set…],      [Calc]>[Make…]>[Arbitrary Set of…]

Very often you need to create a column with special contents for some purpose. You can of course manually enter the data but often it has a certain structure or pattern:
set c41                              # Stores data in c41. From '2' to '10'

2:10/0.1                             # in steps of 0.1.

end

set c45                              # Stores data in c45. From '2' to '10'

4(2:10)                              # in steps of 1, repeated 4 times.

end

set c46                              # Stores data in c45. From '2' to '10' as 2 2 2

4(2:10)3                             # 3 3 3, … 10 10 10, repeated 4 times.

end

set ck11                             # If the constants k11, k41, k45, k46, and k51 are properly
k41(k45:k46)k51                      # explained then the SET-command can be programmed.

end

 Tset
[Calc]>[Make Patterned data]>[Text Values…]

The TSET-command works exactly like the SET-command but is designed for text columns. Here each item must be included in double quotes:
tset c4                              # Stores texts in c4.

"Red" "Blue" "1234"                  # 

end

 Copy
[Data]>[Copy]>[…]
As soon as you analyze set of data you will find that you would like to work with a sub set of the data. There are several ways of doing this and COPY is one way. Learning the details how the command works is best done by using the menus as well as the commands produced in your 'History'-folder shown in your 'Project Manager' (if hidden, find it under your [Windows]-menu of Minitab:
Copy c1-c5 c11-c15;                  # Copies c1-c5 to c11-c15
  Include;                           # including rows where
  Where "c2 > 8 and c4 = 10";        # the condition is true.
  Varnames.                          # Gives the default names to the new columns.

 Concatenate
[Data]>[Concatenate] 

Often you need to connect strings to form a longer string. The example below is perhaps too simple but perhaps you have programmed a static string to be connected to a string created from the data (e.g. a numerical result) to form a part of an output on e.g. a graph. Then you can use CONCATENATE:
Let c4(1) = "red and "             # Stores 'red and ' in first row of c4.

Let c5(1) = "blue"                 # Stores 'blue' in first row of c5.

concatenate c4 c5 c11              # Joins the strings in c4 and c5, result in c11.

 Stat
[Basic Statistics]>[Store Descriptive Statistics]

STAT is an extremely useful command but takes some practice to master. If you issue HELP STAT you will get a long list of information. Suppose that you have data in c1 and some kind of indicator variable in c2 (e.g. supplier as texts or as integer values) and you would like to get various info about c1 divided in its groups:
stat c1;                           # C1 contains your data.

by c2;                             # C2 contains your groupings.

mean c11;                          # Stores (in c11) the mean of each group.

stdev c12.                         # Stores (in c12) the standard deviation of each group..

A trick: suppose that your group column consists of integers and that your goal is to have three columns (mean, stand dev and group number) in c11-c13 for e.g. a special graph. Then you can calculate the mean of the groups and store it in c13 in the following way:
stat c2;                           # Your original group column.

by c2;                             # C2 contains your groupings.

mean c13.                          # Stores (in c11) the mean of each group.

 Any
[Calc]>[Calculator]>[choose Logical]

ANY is a logical command that can be use to manipulate data. Here is an example:
let k1 = 1                         # Stores the value 1 in k1.

let k2 = 12                        # Stores the value 12 in k2.

random 150 c1;                     # Creates 150 integer values in c1

integer 0 20.                      # with the range 0 to 20.

let c3 = ANY(c1; k1; k2)           # ANY value of c1 equal to k1 or k2 becomes 1 in c3. Other wise 0.

 If
[Calc]>[Calculator]>[choose Logical]

IF is also a logical command that can be used to manipulate data. In this example all values less than 7 are coded "male" in c3, all other rows are coded "female":
random 150 c1;                      # Creates 150 integer values in c1

integer 0 20.                       # with the range 0 to 20.

let c3 = IF(c1<7; "male"; "female") # Values less than 7 are coded 'male'. Other wise 'female'.

 If (general)
[Calc]>[Calculator]>[choose Logical]

IF (general) is more general and in this example c5 will contain '12.5' for each satisfying 'c1=3 and c2 > 6'. Also, c5 will contain '16.3' for each row satisfying 'c3 = 7'. All other rows will contain a '*'. (NB that these statements must be designed carefully as there might be conflicts within the command. What is the result in c5 if a row of c1-c3 has the values '3', '9' and '7'?):
random 150 c1-c3;                  # 

integer 2 14.                      # 

let c5 = if(c1 = 3 and c2 > 6; 12.5; c3 = 7; 16.3)
 Tally
[Stat]>[Tables] >[Tally Individual Variables]

TALLY (and its subcommands) is useful when getting to know your data. 'How many suppliers are there?' might be your thinking when looking at a column. Instead of scrolling the column up and down, a TALLY will show you all suppliers, their names and how many rows there are of each one. The list is case sensitive, i.e. the word 'fail' and 'Fail' are two separate expressions (has been useful when checking data that is entered manually:
tally c11                          # Gives a list of unique entries of c11.

tally c11;                         # Stores the output in c12 and c13. If there should be only one 

store c12 c13.                     # value per line the stand dev of c13 can be checked. Should be 0.

 Info
INFO gives a quick answer to what columns you have in your worksheet, what type they are (numeric, text or date), their names, how many rows and if there are any missing rows. This is much faster that to scroll your worksheet to see the e.g. the length of the columns:
Info                               # Gives info about columns, etc, in the current worksheet.

Info c2-c12                        # Gives info about the columns c2-c12.

Info k21-k25                       # Gives info about the constants k21-k25.

 Various text manipulating commands 
Very often you need to manipulate a text column. Perhaps you want to extract a certain number of characters or replace non-printing characters, etc. The menu [Calc]>[Calculator], the [Help]-button and the link Calculator Functions will give you all functions. By clicking any item in the 'Text'-column you will get a detailed list of the possibilities. To master the possibilities you need to read and practice the various items.

The example below: when a SEMC-employee applies for installation of Minitab via ProfileView this lands in our mailbox. Before clicking 'Yes' we copy both 'employee number' and 'name' to be able to offer help, support, up​dates, info, etc. Of course, we could copy number into one column and name into another column but we have all info in one column of a Word document. Together with this list of names some commands are stored. These commands extract the employee number, adds the two characters "; " to each number and then prints the list in the session window. This list is then copied and pasted directly into the address field of Outlook when needed:

Read c1;                             # Reads the following text data into c1.

format(a80).                         # A simple FORMAT, 80 characters wide (80 is maximum).

2884722, Tom, Peng Pung              # Employee number and name.

2345678, Dumbom, Kalle               #   -"-

0675012, Svensson, Karl-Maria        #   -"-

end                                  # Ends the READ.

let c3 = LEFT(c1; 7)                 # Picks the employee number, 7 leftmost characters, into c3.

let k1 = count(c3)                   # Puts number of employee numbers in constant k1.

tset c5                              # Stores k1 rows of the text string '; ' in c5.

k1("; ")                             #

end                                  # Ends the TSET.

conc c3 c5 c7                        # Joins the strings in c3 c5, result in c7.
print c7                             # Prints c7 in the session window. The output is then copied into

                                     # the address filed of Outlook.
 Miscellaneous commands – rmean, rstan, rmax, count, n, nmiss
Often you need to calculate various values across the rows of a number of columns. This is then referred to as 'row statistics'. Use the menu [Calc]>[Row Statistics…] and the [Help]-button to get some information.

There is also often a need to find the number of rows in a column for later use (NB the 'missing value' is design​nated by '*' in numeric and date columns and by an empty cell in text columns):
rmean c1-c50 c51                     # The mean of the data of each row across c1-c50 is stored in c51.

rstan c1-c50 c52                     # The std dev of the row data across c1-c50 is stored in c52.

rmax c1-c50 c53.                     # The maximum of the row data across c1-c50 is stored in c53.

let k101 = count(c234)               # Stores number of rows of c234 in k1.

let k102 = n(c305)                   # Stores number of non-missing values of c305.

let k567 = nmiss(c567)               # Stores number of missing values of c567.

 Some error messages
It is of course impossible to here include all types of errors that you might do. However, here are a few common ones, apart from misspelling a command or general incorrect entries. NB that some messages in your session window are only a message of something unusual or something that you perhaps did not intend and these ones are started by the word 'NOTE'. A real error message is started by 'ERROR': 
Incorrect subcommand.  Many commands consist of several rows therefore a row is terminated by a semicolon to indicate that another row is going to be input. To terminate the input the very last line is ended by a period (a dot, '.') which issues the command. Of course, each subcommand must be a valid one:

random 100 c1 c2                   # Stores 100 data rows in c1 and c2.

plot c1*c2;                        # Starts a plot command.

text 1.2 1.5 'An error';           # Will put a the text 'An error' on the graph.

hist c1 c2                         # Supposed to draw two histograms but will fail as 'hist' will not

                                   # be recognized as it is not a proper subcommand to 'plot'. See

                                   # below.

MTB > random 100 c1 c2             # Stores 100 data rows in c1 and c2.

MTB > plot c1*c2;                  # Starts a plot command.

SUBC> text 1.2 1.5 'An error';

SUBC> hist c1

* ERROR * Subcommand HIST is out of sequence or is not valid with this

        * command.

* ERROR * Subcommand ignored.

SUBC>                           ( Here you must enter a period (dot, '.'), or a correct command!!
Missing semicolon (';').  Sometimes you forget to type the semicolon. The software will note the missing semi​colon but assumes that there is one: 
random 100 c1 c2                   # Stores 100 data rows in c1 and c2.

plot c1*c2;                        # Starts a plot command.

text 1.2 1.5 'An error'            # Will put a the text 'An error' on the graph (note missing ';').

hist c1 c2                         # Supposed to draw two histograms but will fail as 'hist' will not

                                   # be recognized as it is not a proper subcommand to 'plot'. See 

                                   # below.

MTB > random 100 c1 c2             # Stores 100 data rows in c1 and c2.

MTB > plot c1*c2;                  # Starts a plot command.

SUBC> text 1.2 1.5 'An error'      # Puts the text 'An error' on the graph (note missing ';').

* NOTE * Subcommand does not end in . or ; (; assumed).

SUBC>                           ( Here you must enter a period (dot, '.'), or a correct command!!
Incorrect quotation marks.  Some commands use simple quotation marks. Using some word processor might give incorrect marks and thus more or less strange error messages:

name c1 'Result'                   # Names column c1 with the text 'Result'.

